
Revised Oberon-2 is a revision of the programming language Oberon-21. The main difference to 
the original is that it implements a strict superset of Revised Oberon (Oberon-07) as defined in 
2007/20162 rather than being based on the original language Oberon as defined in 1988/19903.   
 
 Original Oberon (1988/1990) ! Original Oberon-2 (1991/1993) 
 
   " 
 

Revised Oberon (2007/2016) ! Revised Oberon-2 (2019/2020) 
 
This document describes only the additions to the Revised Oberon (Oberon-07) programming 
language, namely type-bound procedures, a dynamic heap allocation procedure for fixed-length 
and open arrays, a numeric case statement, exporting and importing of string constants, and no 
access to intermediate objects within nested scopes. For the remaining language features, the 
reader is referred to the official language report of Revised Oberon (Oberon-07)2. 
 
Type-bound procedures 
 
Globally declared procedures may be associated with a record type declared in the same 
module. The procedures are said to be bound to the record type. The binding is expressed by 
the type of the receiver in the heading of a procedure declaration. The receiver may be either a 
variable parameter of record type T or a value parameter of type POINTER TO T (where T is a 
record type). The procedure is bound to the type T and is considered local to it. 
 
     ProcedureHeading = PROCEDURE [Receiver] IdentDef [FormalParameters]. 
     Receiver = "(" [VAR] ident ":" ident ")". 
 
If a procedure P is bound to a type T0, it is implicitly also bound to any type T1 which is an 
extension of T0. However, a procedure P’ (with the same name as P) may be explicitly bound to 
T1 in which case it overrides the binding of P. P’ is considered a redefinition of P for T1. The 
formal parameters of P and P’ must match. If P and T1 are exported, P’ must be exported too. 
 
If v is a designator and P is a type-bound procedure, then v.P denotes that procedure P which is 
bound to the dynamic type of v. This may be a different procedure than the one bound to the 
static type of v. v is passed to P's receiver according to the standard parameter passing rules. 
 
If r is a receiver parameter declared with type T, r.P^ (pronounced r.P-referenced) denotes the 
(redefined) procedure P bound to the base type of T. 
 
Example: 
																																																								
1 Mössenböck H., Wirth N.: The Programming Language Oberon-2. Structured Programming, 12(4):179-195, 1991 
2 http://www.inf.ethz.ch/personal/wirth/Oberon/Oberon07.Report.pdf (Revision 3.5.2016) 
3 https://inf.ethz.ch/personal/wirth/Oberon/Oberon.Report.pdf  (Revision 1.10.1990) 

MODULE Trees; 
  IMPORT Out; 
 
  TYPE Tree = POINTER TO Node; 
    Node = RECORD key : INTEGER; 
      left, right: Tree 
    END ; 
 
    CenterTree = POINTER TO CenterNode; 
    CenterNode = RECORD (Node) width: INTEGER; 
      subnode: Tree 
    END ; 
 
  PROCEDURE (T: Tree) Insert (node: Tree);               (*procedure bound to Tree*) 
    VAR p, father: Tree; 
  BEGIN p := T; 
    REPEAT father := p; 
      IF node.key < p.key THEN p := p.left 
      ELSIF node.key > p.key THEN p := p.right 
      ELSE p := NIL 
      END 
    UNTIL p = NIL; 
    IF node.key < father.key THEN father.left := node ELSE father.right := node END ; 
    node.left := NIL; node.right := NIL 
  END Insert; 
 
  PROCEDURE (T: CenterTree) Insert (node: Tree);    (*redefinition of Insert bound to CenterTree*) 
  BEGIN Out.Int(node(CenterTree).width, 3); 
    T.Insert^(node)                                                          (*calls the Insert procedure bound to Tree*) 
  END Insert; 
END Trees. 

   
Dynamic heap allocation procedure for fixed-length and open arrays 
 
If p is a variable of type P = POINTER TO T, a call of the predefined procedure NEW allocates 
a variable of type T in free storage at run time. The type T can be a record or array type. 
 
If T is a record type or an array type with fixed length, the allocation has to be done with 
 

NEW(p) 
 
If T is an open array type, the allocation has to be done with 
 

NEW(p, len) 
 
where T is allocated with the length given by the expression len, which must be an integer type. 
 
In either case, a pointer to the allocated variable is assigned to p. This pointer p is of type P, 
while the referenced variable p^ (pronounced p-referenced) is of type T. 
 
If T is a record type, a field f of an allocated record p^ can be accessed as p^.f or as p.f. If T is 
an array type, the elements of an allocated array p^ can be accessed as p^[0] to p^[len-1] or as 
p[0] to p[len-1], i.e. record and array selectors imply dereferencing. 
 
If T is an array type, its element type can be a record, pointer, procedure or a basic type (BYTE, 
BOOLEAN, CHAR, INTEGER, REAL, SET), but not an array type (no multi-dimensional arrays). 

Example: 

MODULE Test; 
  TYPE R = RECORD x, y: INTEGER END ; 
    A = ARRAY OF R;     (*open array*) 
    B = ARRAY 20 OF INTEGER;   (*fixed-length array*) 
    P = POINTER TO A;      (*pointer to open array*) 
    Q = POINTER TO B;      (*pointer to fixed-length array*) 
 
  VAR a: P; b: Q; 
 
  PROCEDURE New1*; 
  BEGIN NEW(a, 100);  a[53].x := 1 
  END New1; 
 
  PROCEDURE New2*; 
  BEGIN NEW(b);  b[3] := 2 
  END New2; 
END Test. 

 
The following rules and restrictions apply4: 
 
• Bounds checks on fixed-length arrays are performed at compile time. 
• Bounds checks on open arrays are performed at run time. 
• If P is of type P = POINTER TO T, the type T must be a named record or array type5. 
 
Numeric case statement 
 
The revised compiler brings the compiler in line with the official Oberon-07 language report, and 
now also allows numeric case statements6 in addition to type case statements. 
 
Case statements specify the selection and execution of a statement sequence according to the 
value of an expression. First the case expression is evaluated, then the statement sequence is 
executed whose case label list contains the obtained value. 
 
     CaseStatement =   CASE expression OF case {"|" case} [ELSE StatementSequence] END. 
     case  =   [CaseLabelList ":" StatementSequence]. 
     CaseLabelList =   LabelRange {"," LabelRange}.  
     LabelRange =   label [".." label]. 
     label  =   integer | string | qualident. 
 
Numeric case statements 
 
If the case expression is of type INTEGER or CHAR, all case labels must be integers or single-
character strings, respectively. 
 
Example: 
 

																																																								
4 Note that allocating dynamic arrays requires a modified version of the inner core module Kernel, which introduces a new kind of heap block –  a r r a y  block in addition to  r e c o r d  block (in 
some implementations of the Oberon system, an additional kind of heap block describing a storage block of n bytes (“sysblk”) exists, which is typically allocated by a special low-level procedure 
SYSTEM.NEW(p, n). In our implementation, no such special procedure is necessary, as it is covered by a call to NEW(p, n), where p is a pointer to an array of BYTE). Array blocks allocated 
using NEW(p, len) or NEW(p) are garbage-collected in the same way as regular record blocks. The implementation of garbage collection on fixed-length and open arrays is similar to other 
implementations of the Oberon system. See, for example, “Oberon Technical Notes: Garbage collection on open arrays", J. Templ, ETH technical report, March 1991. 
5 Restricting pointers to  n a m e d  arrays is consistent with the official Oberon-07 compiler, which restricts pointers to point to  n a m e d  records. 
6 http://github.com/andreaspirklbauer/Oberon-numeric-case-statement 

CASE k OF 
    | 0: x := x + y 
    | 1: x := x − y 
    | 2: x := x * y 
    | 3: x := x / y 
END 

 
Type case statements 
 
The type T of the case expression (case variable) may also be a record or pointer type. Then 
the case labels must be extensions of T, and in the statements Si labelled by Ti, the case 
variable is considered as of type Ti. 
 
Example: 

TYPE R   = RECORD a: INTEGER END ; 
    R0 = RECORD (R) b: INTEGER END ; 
    R1 = RECORD (R) b: REAL END ; 
    R2 = RECORD (R) b: SET END ; 
    P   = POINTER TO R; 
    P0 = POINTER TO R0; 
    P1 = POINTER TO R1; 
    P2 = POINTER TO R2; 
 
VAR p: P; 
 
CASE p OF 
    | P0: p.b := 10 
    | P1: p.b := 2.5 
    | P2: p.b := {0, 2} 
END 

 
The following rules and restrictions apply: 
 
• Case labels of numeric case statements must have values between 0 and 255. 
• Case variables of type case statements must be simple identifiers that cannot be followed by 

selectors, i.e. they cannot be elements of a structure (array elements or record fields). 
• If the value of the case expression does not correspond to any case label in the source text, 

the statement sequence following the symbol ELSE is selected, if there is one, otherwise no 
action is taken (in the case of a type case statement) or the program is aborted (in the case 
of a numeric case statement)7. 

 
The ELSE clause has been re-introduced even though it is not part of the Oberon-07 language 
definition. This was done mainly for backward compatibility reasons. In general, we recommend 
using the ELSE clause only in well-justified cases, for example if the index range far exceeds 
the label range. But even in that case, one should first try to find a representation using explicit 
case label ranges, as shown in the example below (which assumes an index range of 0..255). 
 

 CASE i OF 
  | 1:  S1 
  | 3:  S3 
  | 7:  S7 
  | 9:  S9 
ELSE S0 
END 

 
 
is the same as 

CASE i OF 
  | 1:  S1 
  | 3:  S3 
  | 7:  S7 
  | 9:  S9 
  | 0, 2, 4..6, 8, 10..255:  S0          (*preferred*) 
END 

																																																								
7 If one wants to treat such events as “empty” actions, an empty ELSE clause can be used. 

Exporting and importing of string constants 
 
The revised compiler brings the compiler in line with the official Oberon-07 language report, and 
now allows exporting and importing of string constants8. Exported string constants are treated 
like pre-initialized, immutable exported variables. 
 
Example: 

MODULE M; 
  CONST str* = "This is a sample string";               (*exported string constant*) 
END M. 
 
MODULE N; 
  IMPORT M, Out; 
 
  PROCEDURE P*; 
  BEGIN Out.Str(M.str)                                           (*print the imported string constant*) 
  END P; 
END N. 

 
No access to intermediate objects within nested scopes 
 
The revised compiler brings the compiler in line with the official Oberon-07 language report, and 
now also disallows access to intermediate constants and types within nested scopes, not just 
access to intermediate variables9. 
 
Like the official Oberon-07 compiler, the revised compiler implements shadowing through scope 
when accessing named objects. This means when two objects share the same name, the one 
declared at the narrower scope hides, or shadows, the one declared at the wider scope. In such 
a situation, the shadowed element is not available in the narrower scope. If the shadowing 
element is itself declared at an intermediate scope, it is only available at that scope level, but 
not in narrower scopes (as access to intermediate objects is disallowed). 
 
The official Oberon-07 compiler already issues an error message, if intermediate variables are 
accessed within nested scopes (line 25 of the program below), regardless of whether a global 
variable with the same name exists (line 7) or not. With the revised compiler, the same error 
message is now also issued for intermediate constants (line 21) and types (lines 16 and 18). 
 
Example: 

MODULE Test; 
  CONST C = 10;  (*global constant C, shadowed in Q and therefore not available in R*) 
 
  TYPE G = REAL;  (*global type G, not shadowed in Q and therefore available in R*) 
      T = REAL;   (*global type T, shadowed in Q and therefore not available in R*) 
  VAR A,   (*global variable A, not shadowed in Q and therefore available in R*) 
      B: INTEGER;  (*global variable B, shadowed in Q and therefore not available in R*) 
 
  PROCEDURE P;  (*global procedure P*) 
  
    PROCEDURE Q;  (*intermediate procedure Q, contains shadowing elements C, T and B*) 

																																																								
8 http://github.com/andreaspirklbauer/Oberon-importing-string-constants 
9 http://github.com/andreaspirklbauer/Oberon-no-access-to-intermediate-objects 

      CONST C = 20;  (*intermediate constant C which shadows the global constant C*) 
      TYPE T = INTEGER; (*intermediate type T which shadows the global type T*) 
      VAR B: INTEGER;  (*intermediate variable B which shadows the global variable B*) 
 
      PROCEDURE R(x: T): T; (*access to intermediate type T allowed in original, not allowed in modified compiler*) 
        VAR i: INTEGER; 
          q: T;   (*access to intermediate type T allowed in original, not allowed in modified compiler*) 
          g: G;   (*access to global type G (not shadowed) allowed in both compilers*) 
      BEGIN (*R*) 
        i := C;   (*access to interm. constant C allowed in original, not allowed in modified compiler*) 
        P;    (*access to global (unshadowed) procedure P allowed in both compilers*) 
        Q;    (*access to intermediate procedure Q allowed in both compilers*) 
        i := A;   (*access to global (unshadowed) variable A allowed in both compilers*) 
        i := B;   (*access to intermediate variable B not allowed in both compilers*) 
        RETURN i 
      END R; 
    END Q; 
  END P; 
 
END Test. 

 
Disallowing access to intermediate objects from within nested scopes while at the same time 
implementing shadowing through scope raises the question whether one should relax the 
shadowing rules and allow access to the global scope level, when an object with the same 
name as a global object is re-declared at an intermediate level, but not at the strictly local level 
(“piercing through the shadow"). 
 
In the above example, such an approach would allow access to the global variable B (line 7) in 
procedure R (line 25), effectively ignoring any intermediate-level variables B that may also exist 
(line 14). It would make nested procedures “self-contained” in the sense that they can be moved 
around freely. For example, procedure R can be made local to procedure Q without having to 
be concerned about whether one can still access the global variable B (line 7). 
 
We have opted not to adopt this approach for two main reasons. First, a nested procedure may 
also call the surrounding procedure that contains it (a frequent case) and is thus not necessarily 
self-contained anyway. Second, we didn’t want to break with a long language tradition10. 
 
 
 
 
 
 

* * * 
 

																																																								
10 In the appendix of http://github.com/andreaspirklbauer/Oberon-no-access-to-intermediate-objects, a possible implementation of such relaxed shadowing rules is provided. 

Appendix: Implementation cost of the Revised Oberon-2 language additions 
 
The total aggregate implementation cost of the Revised Oberon-2 language additions in source 
lines of code (sloc) is as follows11: 
 

Compiler module Revised Oberon (Oberon-07) Revised Oberon-2 Difference Percent 
ORS (scanner) 293 293 0 0 % 
ORB (base) 394 452 58 + 14.7 % 
ORG (generator) 984 1102 118 + 12.0 % 
ORP (parser) 949 1093 144 + 15.2 % 
Total 2620 2940 320 + 12.2 % 

 
Feature Source lines of code 
Type-bound procedures 200 
Dynamic heap allocation procedure for fixed-length and open arrays 40 
Numeric case statement 65 
All other features combined 15 
Total 320 

 
 
 
 
 
 

* * * 
 

																																																								
11 Not counting empty lines and about 100-150 additional lines of source code in modules Kernel, Modules and System to complement the implementation of the Revised Oberon-2 language. 


